Вычисление пределов, производных, исследование функции, разложение в ряд и аппроксимация

1. Вычисление пределов.
2. Дифференцирование.
3. Исследование функции одной переменной, нахождение точек разрывов и экстремумов.
4. Экстремумы функции многих переменных.
5. Разложение в ряд и аппроксимация.

Всюду далее примеры работы пакета Maple будут приведены в режиме интерфейса Worksheet Mode с текстовым режимом ввода команд и выражений (Text Mode). При выполнении заданий и упражнений используйте режим интерфейса Worksheet Mode с режимом ввода «двумерной математики» Maple Math Mode или ввода в строку Text Mode.

В *Maple* для некоторых математических операций существует по две команды: одна прямого, а другая – отложенного исполнения. Имена команд состоят из одинаковых букв за исключением первой: команды прямого исполнения начинаются со строчной буквы, а команды отложенного исполнения – с заглавной. После обращения к команде отложенного действия математические операции (интеграл, предел, производная и т.д.) выводятся на экран в виде стандартной аналитической записи этой операции. Вычисление в этом случае сразу не производится. Команда прямого исполнения выдает результат сразу.

§1. Вычисление пределов

Для вычисления пределов имеются две команды:
1) прямого исполнения – **`limit(expr,x=a,par)`**, где **`expr`** – выражение, предел которого следует найти, **`a`** – значение точки, для которой вычисляется предел, **`par`** – необязательный параметр для поиска односторонних пределов (**`left`** – слева, **`right`** – справа) или указание типа переменной (**`real`** – действительная, **`complex`** – комплексная).
Команду прямого исполнения **`limit`** можно также записать с помощью шаблона $\lim_{x \to a} f$ на палитре Expressions.
2) отложенного исполнения – **`Limit(expr,x=a,par)`**, где параметры команды такие же, как и в предыдущем случае. Пример действий этих команд:

```
> Limit(sin(2*x)/x,x=0);
```

$$\lim_{x \to 0} \frac{\sin(2x)}{x}$$

```
> limit(sin(2*x)/x,x=0);
```

$$2$$

С помощью этих двух команд можно записывать математические выкладки в стандартном аналитическом виде, например:

```
> Limit(x*(Pi/2+arctan(x)),x=-infinity)=
```

```
limit(x*(Pi/2+arctan(x)), x=-infinity);
```

$$\lim_{x \to -\infty} x \left(\frac{1}{2} \pi + \arctan(x) \right) = -1$$

Односторонние пределы вычисляются с указанием параметров: **left** – для нахождения предела слева и **righ** – справа. Например:

```
> Limit(1/(1+exp(1/x)),x=0,left)=
  limit(1/(1+exp(1/x)),x=0,left);
```

$$\lim_{x \to 0-} \frac{1}{1+e^{\frac{1}{x}}} = 1$$

```
> Limit(1/(1+exp(1/x)),x=0,right)=
  limit(1/(1+exp(1/x)), x=0,right);
```

$$\lim_{x \to 0+} \frac{1}{1+e^{\frac{1}{x}}} = 0$$

Задание 1.

1. Вычислить предел $\lim\limits_{x \to 1} (1-x)\text{tg}\dfrac{\pi x}{2}$. Наберите:

```
> Limit((1-x)*tan(Pi*x/2),x=1)=
  limit((1-x)*tan(Pi*x/2),x=1);
```

$$\lim_{x \to 1} (1-x) \tan\left(\frac{1}{2} \pi x \right) = \frac{2}{\pi}$$

2. Найти односторонние пределы $\lim\limits_{x \to 1-} \text{arctg}\dfrac{1}{1-x}$ и $\lim\limits_{x \to 1+} \text{arctg}\dfrac{1}{1-x}$. Наберите:

```
> Limit(arctan(1/(1-x)),x=1,left)=
  limit(arctan(1/(1-x)), x=1, left);
```

$$\lim_{x \to 1^-} \arctan\left(\frac{1}{1-x} \right) = \frac{1}{2} \pi$$

```
> Limit(arctan(1/(1-x)),x=1,right)=
  limit(arctan(1/(1-x)),x=1, right);
```

$$\lim_{x \to 1+} \arctan\left(\frac{1}{1-x} \right) = -\frac{1}{2} \pi$$

§2. Дифференцирование

Вычисление производных.

Для вычисления производных в *Maple* имеются две команды:

1) прямого исполнения – **diff(f,x)**, где **f** – функция, которую следует продифференцировать, **x** – имя переменной, по которой производится дифференцирование. Аналогом команды **diff(f,x)** является шаблон $\frac{d}{dx} f$ на палитре Expressions.

2) отложенного исполнения – **Diff(f,x)**, где параметры команды такие же, как и в предыдущей. Действие этой команды сводится к аналитической записи производной в

виде $\frac{\partial}{\partial x} f(x)$. После выполнения дифференцирования, полученное выражение желательно упростить. Для этого следует использовать команды **simplify factor** или **expand**, в зависимости от того, в каком виде вам нужен результат.

Пример:

```
> Diff(sin(x^2),x)=diff(sin(x^2),x);
```

$$\frac{d}{dx} \sin(x^2) = 2\cos(x^2)\, x$$

Для вычисления производных старших порядков следует указать в параметрах **x$n**, где **n** – порядок производной; например:

```
> Diff(cos(2*x)^2,x$4)=diff(cos(2*x)^2,x$4);
```

$$\frac{d^4}{dx^4} \left(\cos(2\,x)^2\right) = -128\sin(2\,x)^2 + 128\cos(2\,x)^2$$

Полученное выражение можно упростить двумя способами:

```
> simplify(%);
```

$$\frac{d^4}{dx^4} \left(\cos(2\,x)^2\right) = 256\cos(2\,x)^2 - 128$$

```
> combine(rhs(%%));
```

$$128\cos(4\,x)$$

Частные производные.

Для вычисления частных производных функции $f(x_1,\ldots, x_m)$ используется та же команда **diff**. В этом случае эта команда имеет такой формат: **diff(f,x1$n1,x2$n2,…, xm$nm)**, где **x1,…, xm** – переменные, по которым производится дифференцирование, а после знака **$** указаны соответствующие порядки дифференцирования. Например, частная производная $\frac{\partial^2 f}{\partial x \partial y}$ записывается в виде: **diff(f,x,y).**

Для вычисления частных производных можно использовать шаблон $\frac{\partial}{\partial x} f$ на палитре Expressions (команда прямого исполнения **diff(f,x)**).

Дифференциальный оператор.

Для определения дифференциального оператора используется команда **D(f)** – **f**- функция. Например:

```
> D(sin);
```

$$\cos$$

Вычисление производной в точке:

```
> D(sin)(Pi):eval(%);
```

$$-1$$

Оператор дифференцирования применяется к функциональным операторам

```
> f:=x-> ln(x^2)+exp(3*x):
> D(f);
```

$$x \to \frac{2}{x} + 3\,e^{3x}$$

Задание 2.

1. Вычислить производную $f(x) = \sin^3 2x - \cos^3 2x$

```
> Diff(sin(2*x)^3-cos(2*x)^3,x)=
diff(sin(2*x)^3-cos(2*x)^3,x);
```

$$\frac{d}{dx}\left(\sin(2x)^3 - \cos(2x)^3\right)$$
$$= 6\sin(2x)^2\cos(2x)$$
$$+ 6\cos(2x)^2\sin(2x)$$

2. Вычислить $\frac{d^{24}}{dx^{24}}(e^x(x^2-1))$. Наберите:

```
> Diff(exp(x)*(x^2-1),x$24)=
diff(exp(x)*(x^2-1),x$24);
```

$$\frac{d^{24}}{dx^{24}}\left(e^x\left(x^2-1\right)\right) = e^x\left(x^2-1\right)$$
$$+ 48\,e^x x + 552\,e^x$$

```
> collect(%,exp(x));
```

$$\frac{d^{24}}{dx^{24}}\left(e^x\left(x^2-1\right)\right) = \left(x^2 + 551\right.$$
$$\left.+ 48\,x\right)e^x$$

3. Вычислить вторую производную функции $y = \sin^2 x/(2+\sin x)$ в точках $x=\pi/2$, $x=\pi$.

```
> y:=sin(x)^2/(2+sin(x)): d2:=diff(y,x$2):
> d2y(Pi)=eval(d2, x = Pi);
```

$$d2y(\pi) = 1$$

```
> d2y(Pi/2)=eval(d2, x = Pi);
```

$$d2y\left(\frac{1}{2}\pi\right) = 1$$

4. Найти $\frac{\partial f}{\partial x}$ и $\frac{\partial f}{\partial y}$ функции $f = \operatorname{arctg}\frac{x}{y}$.

```
> f:=arctan(x/y):
> Diff(f,x)=simplify(diff(f,x));
```

$$\frac{\partial}{\partial x}\arctan\frac{x}{y} = \frac{y}{x^2+y^2}$$

```
> Diff(f,y)=simplify(diff(f,y));
```

$$\frac{\partial}{\partial y}\arctan\frac{x}{y} = -\frac{x}{x^2+y^2}.$$

5. Найти все частные производные 2-го порядка функции $f(x,y) = \frac{x-y}{x+y}$.

```
> restart; f:=(x-y)/(x+y):
> Diff(f,x$2)=simplify(diff(f,x$2));
```

$$\frac{\partial^2}{\partial x^2}\frac{x-y}{x+y} = -4\frac{y}{(x+y)^3}$$

> **Diff(f,y$2)=simplify(diff(f,y$2));**

$$\frac{\partial^2}{\partial y^2}\frac{x-y}{x+y}=4\frac{x}{(x+y)^3}$$

> **Diff(f,x,y)=diff(f,x,y);**

$$\frac{\partial^2}{\partial x\partial y}\frac{x-y}{x+y}=2\frac{x-y}{(x+y)^3}.$$

§3. Исследование функции одной переменной

Исследование функции необходимо начинать с нахождения ее области определения, но, к сожалению, это трудно автоматизируемая операция. Поэтому при рассмотрении этого вопроса приходится решать неравенства. Однако, ответить на вопрос, определена ли функция на всей числовой оси, или нет, можно исследовав ее на непрерывность.

Непрерывность функции и точки разрыва.

Проверить непрерывность функции $f(x)$ на заданном промежутке $[x_1,x_2]$ можно с помощью команды **`iscont(f,x=x1..x2)`**. Если функция **`f`** непрерывна на этом интервале, то в поле вывода появится ответ *true* – (истина); если функция **`f`** не является непрерывной на этом интервале, то в поле вывода появится ответ *false* – (ложь). В частности, если задать интервал **`x=-infinity..+infinity`**, то функция **`f`** будет проверяться на всей числовой оси. В этом случае, если будет получен ответ *true,* то можно сказать, что функция определена и непрерывна на всей числовой оси. В противном случае следует искать точки разрыва. Это можно сделать двумя способами:

1) с помощью команды **`discont(f,x)`**, где **`f`** – функция, исследуемая на непрерывность, **`x`** – переменная. Эта команда пригодна для нахождения точки разрыва первого и второго родов.
2) с помощью команды **`singular(f,x)`**, где **`f`** – функция, **`x`** – переменная. Эта команда годится для нахождения точек разрыва второго рода как для вещественных значений переменной, так и для комплексных.

Обе эти команды выдают результаты в виде перечисления точек разрыва в фигурных скобках (в виде множества, *set*).

Задание 3.1.

1. Найдите точки разрыва функции $y=e^{\frac{1}{x+3}}$

   > **iscont(exp(1/(x+3)),x=-infinity..+infinity);**

   false

   Это означает, что функция не является непрерывной. Поэтому следует найти точки разрыва с помощью команды:

   > **discont(exp(1/(x+3)),x);**

   {-3}

   Ответ наберите в текстовом режиме в новой строке:

   “Точка разрыва x=–3.”

2. Найти точки разрыва функции $y = \operatorname{tg}\frac{x}{2-x}$

> **iscont(tan(x/(2-x)),x=-infinity..infinity);**

$$false$$

> **singular(tan(x/(2-x)),x);**

$$\{x=2\},\ \left\{x = \frac{2\pi\,(2\ \_Z1\sim + 1)}{-2 + 2\ \_Z1\sim\pi + \pi}\right\}$$

Здесь *_Z1* – целые числа. Ответ наберите в текстовом режиме в новой строке: “Точки разрыва: x=2 и x=2π(2n+1)/(π(2n+1)-2).”

Экстремумы. Наибольшее и наименьшее значение функции.

В *Maple* для исследования функции на экстремум имеется команда **extrema(f,{cond},x,'s')** , где **f** - функция, экстремумы которой ищутся, в фигурных скобках **{cond}** указываются ограничения для переменной, **x** – имя переменной, по которой ищется экстремум, в апострофах **'s'** – указывается имя переменной, которой будет присвоена координата точки экстремума. Если оставить пустыми фигурные скобки {}, то поиск экстремумов будет производиться на всей числовой оси. Результат действия этой команды относится к типу *set*. Пример:

> **extrema(arctan(x)-ln(1+x^2)/2,{},x,'x0');x0;**

$$\{\frac{\pi}{4} - \frac{1}{2}\ln(2)\}$$

$$\{\{x=1\}\}$$

В первой строке вывода приводится экстремум функции, а во второй строке вывода – точка этого экстремума.

К сожалению, эта команда не может дать ответ на вопрос, какая из точек экстремума есть максимум, а какая – минимум. Для нахождения максимума функции $f(x)$ по переменной x на интервале $x \in [x1, x2]$ используется команда **maximize(f,x,x=x1..x2),** а для нахождения минимума функции $f(x)$ по переменной x на интервале $x \in [x1, x2]$ используется команда **minimize(f, x, x=x1..x2)**. Если после переменной указать **'infinity'** или интервал **x=-infinity..+infinity**, то команды **maximize** и **minimize** будут искать, соответственно, максимумы и минимумы на всей числовой оси как во множестве вещественных чисел, так и комплексных. Если такие параметры не указывать, то поиск максимумов и минимумов будет производиться только во множестве вещественных чисел. Пример:

> **maximize(exp(-x^2),x);**

$$1$$

Недостаток этих команд в том, что они выдают только значения функции в точках максимума и минимума, соответственно. Поэтому для того, чтобы полностью решить задачу об исследовании функции $y=f(x)$ на экстремумы с указанием их характера (max или min) и координат (x, y) следует сначала выполнить команду:

> **extrema(f,{},x,'s');s;**

а затем выполнить команды **maximize(f,x); minimize(f,x).** После этого будут полностью найдены координаты всех экстремумов и определены их характеры (max или min).

Команды **maximize** и **minimize** быстро находят абсолютные экстремумы, но не всегда пригодны для нахождения локальных экстремумов. Команда **extrema** вычисляет также критические точки, в которых функция не имеет экстремума. В этом случае экстремальных значений функции в первой строке вывода будет меньше, чем вычисленных критических точек во второй строке вывода. Выяснить характер найденного экстремума функции $f(x)$ в точке $x=x_0$ можно, если вычислить вторую производную в этой точке и по ее знаку сделать вывод: если $f''(x_0)>0$, то в точке x_0 будет min, а если $f''(x_0)<0$ – то max.

Координаты точек максимума или минимума можно получить, если в параметрах команд **maximize** и **minimize** после переменной записать через запятую новую опцию **location**. В результате в строке вывода после самого максимума (минимума) функции будут в фигурных скобках указаны координаты точек максимума (минимума). Например:

```
> minimize(x^4-x^2, x, location);
```

$$\frac{-1}{4},\ \left\{\left[\left\{x=-\frac{1}{2}\sqrt{2}\right\}, -\frac{1}{4}\right], \left[\left\{x=\frac{1}{2}\sqrt{2}\right\}, -\frac{1}{4}\right]\right\}$$

В строке вывода получились координаты минимумов и значения функции в этих точках.

Задание 3.2.

1. Найти max и min $y=\frac{1}{2}(x^2-\frac{1}{2})\arcsin x+\frac{x}{4}\sqrt{1-x^2}-\frac{\pi}{12}x^2$.

```
> y:=(x^2-1/2)*arcsin(x)/2+x*sqrt(1-x^2)/4-Pi*x^2/12:
> extrema(y, {}, x, 's'); s;
```

$$\left\{0, -\frac{1}{24}\pi+\frac{1}{16}\sqrt{3}\right\}$$

$$\left\{\{x=0\}, \left\{x=\frac{1}{2}\right\}\right\}$$

После выполнения этих команд найдены экстремумы функции и точки экстремумов. Порядок следования x–координат экстремумов во второй строке вывода соответствует порядку следования значений экстремумов в первой строке вывода. Таким образом, найдены экстремумы в точках (0,0) и (1/2, $-\pi/24+\sqrt{3}/16$). Осталось выяснить, какая из них является максимумом, а какая – минимумом. Для этого используйте команды **maximize** и **minimize**. Без указания интервала по x команды не вычисляют экстремумы (отметим, что область определения фукции y: $D(y)=(-1,1)$.

```
> minimize(y);
Error, (in minimize/cell/univariate) cannot minimize over a
complex valued function
> maximize(y);
Error, (in maximize) cannot minimize over a complex valued
function
> ymax := maximize(y, x = -1 .. 1/2,location);
```

$$ymax := 0,\ \{[\{x=0\}, 0]\}$$

```
> ymin := minimize(y, x = 1/2 .. 1, location);simplify(ymin[1]);
```

$$ymin := -\frac{1}{24}\pi + \frac{1}{32}\sqrt{3}\sqrt{4},\ \left\{\left[\left\{x = \frac{1}{2}\right\}, -\frac{1}{24}\pi + \frac{1}{32}\sqrt{3}\sqrt{4}\right]\right\}$$

$$-\frac{1}{24}\pi + \frac{1}{16}\sqrt{3}$$

Ответ наберите в текстовом режиме в новой строке:

“Экстремумы: $\max y(x) = y(0) = 0$, $\min y(x) = y(1/2) = -\pi/24 + \sqrt{3}/16$.”

Для набора математических символов и греческих букв в текстовом режиме можно использовать соответствующие шаблоны. Для возвращения в текстовый режим снова следует нажать на кнопку с буквой «**T**».

2. Найдите наибольшее и наименьшее значение $f(x) = x^2 \ln x$ на интервале $x \in [1,2]$. Наберите:

```
> f:=x^2*ln(x):
> fmax:=maximize(f,x,x=1..2,location);evalf(fmax);
> fmin:=minimize(f,x,x=1..2,location);
> plot(f,x=0..2,-1..5);
```

$$fmax := 4\ln(2),\ \{[\{x = 2\}, 4\ln(2)]\}$$

$$2.772588722,\ \{[\{x = 2.\}, 2.772588722]\}$$

$$fmin := 0,\ \{[\{x = 1\}, 0]\}$$



Ответ наберите в текстовом режиме в новой строке:

”Наибольшее значение: $\max f(x) = 4\ln 2$, наименьшее значение $\min f(x) = 0$“. Таким образом, минимум и максимум достигаются на концах отрезка [1,2].

3. Найти экстремумы функции $y = \dfrac{x^3}{4 - x^2}$ и установить их характер с помощью второй производной. Наберите:

```
>restart:y:=x^3/(4-x^2):
>extrema(y,{},x,'s');s;
```

$$\{-3\sqrt{3}, 3\sqrt{3}\}$$

$$\{\{x = 0\}, \{x = -2\sqrt{3}\}, \{x = 2\sqrt{3}\}\}$$

Получено два экстремума и три критические точки. Исследование можно продолжить с помощью второй производной:

```
>d2:=diff(y,x$2): d2y(x)=eval(d2,x=0);
```

$$d2y(x) = 0$$

```
> d2y(-2*sqrt(3))=eval(d2,x=-2*sqrt(3));
```

$$d2y(-2\sqrt{3}) = \frac{3}{4}\sqrt{3}$$

```
> d2y(2*sqrt(3))=eval(d2,x=2*sqrt(3));
```

$$d2y(2\sqrt{3}) = -\frac{3}{4}\sqrt{3}$$

Так как $y''(0)=0$, то в точке x=0 нет экстремума; так как $y''(2\sqrt{3})<0$, то в точке $x=2\sqrt{3}$ будет max; так как $y''(-2\sqrt{3})>0$, то в точке $x=-2\sqrt{3}$ будет min. Перейдите в текстовый режим и запишите ответ в виде:

“Максимум в точке ($2\sqrt{3}, 3\sqrt{3}$), минимум в точке ($-2\sqrt{3}, -3\sqrt{3}$)”.

Исследование функции по общей схеме

1. Область определения функции

Область определения функции $f(x)$ – полностью может быть указана после исследования функции на непрерывность.

2. Непрерывность и точки разрыва

Непрерывность и точки разрыва функции $f(x)$ исследуются по схеме:

```
> iscont(f, x=-infinity..infinity);
> d1:=discont(f,x);
> d2:=singular(f,x);
```

В результате наборам переменным **d1**и **d2** будут присвоены значения x-координат в точках разрыва 1 и 2-го родов (если они будут найдены).

3. Исследование функции на четность и нечетность.

Для определения четности функции (f(−x)=f(x)) служит команда **type(f, evenfunc(x))**, а установить нечетность функции (f(-x)=−f(x)) можно с помощью команды **type(f, oddfunc(x))**

4. Исследование функции на периодичность.

f(x)=f(x+T)

5. **Асимптоты**.

Точки бесконечных разрывов определяют графика $f(x)$. Уравнение **вертикальной асимптоты** имеет вид:

```
> xr:=d2;
```

Поведение функции $f(x)$ на бесконечности характеризуется **наклонными асимптотами** (если они есть). Уравнение наклонной асимптоты $y_i=k_ix+b_i$, i=1,2, где коэффициенты вычисляются по формулам:

$$k_1 = \lim_{x\to+\infty} \frac{f(x)}{x} \text{ и } b_1 = \lim_{x\to+\infty} (f(x) - k_1 x),$$

$$k_2 = \lim_{x\to-\infty} \frac{f(x)}{x} \text{ и } b_2 = \lim_{x\to-\infty} (f(x) - k_2 x).$$

Аналогичные формулы для $x\to-\infty$. Поэтому нахождение наклонных асимптот можно провести по следующей схеме:

```
> k1:=limit(f(x)/x, x=+infinity);
> b1:=limit(f(x)-k1*x, x=+infinity);
> k2:=limit(f(x)/x, x=-infinity);
> b2:=limit(f(x)-k2*x, x=-infinity);
```

Часто оказывается, что **k1=k2** и **b1=b2**, в этом случае будет одна асимптота при $x\to+\infty$ и при $x\to-\infty$. С учетом этого составляется уравнение асимптоты

> **yn:=k1*x+b1;**

6. Экстремумы, их характер, значения функции в точках экстремума.

Исследование функции $f(x)$ на экстремумы можно проводить по схеме:

> **extrema(f(x), {}, x, ’s’);**

> **s;**

> **fmax:=maximize(f(x), x);**

> **fmin:=minimize(f(x), x);**

После выполнения этих команд будут найдены координаты (x, y) всех максимумов и минимумов функции $f(x)$.

7. Построение графика функции с указанием координат экстремумов.

Построение графика функции $f(x)$ – это окончательный этап исследования функции. На рисунке помимо графика исследуемой функции $f(x)$ должны быть нанесены все ее асимптоты пунктирными линиями, подписаны координаты точек max и min. Приемы построения графиков нескольких функций и нанесения надписей были рассмотрены в теме III.

Задание 3.3.

1. Провести полное исследование функции $f(x)=\dfrac{x^4}{(1+x)^3}$ по общей схеме. Сначала перейдите в текстовый режим и наберите “Исследование функции: “. Затем вернитесь в режим командной строки и наберите команды:

> **f:=x^4/(1+x)^3:**

В текстовом режиме наберите “Непрерывность функции”. В режиме командной строки и наберите:

> **iscont(f, x=-infinity..infinity);**

$$false$$

Это означает, что функция не является непрерывной. Перейдите в текстовый режим и наберите “Нахождение точек разрыва”. Вернитесь в режим командной строки и наберите:

> **discont(f,x);**

$$\{-1\}$$

Присвоим полученное значение точки разрыва переменной **xr**:

> **xr:= %[1];**

$$xr:=-1$$

Перейдите в текстовый режим и наберите: “Получена точка бесконечного разрыва x=–1”. С новой строки наберите: “Нахождение асимптот.”. Перейдите на новую строку и наберите “Уравнение вертикальной асимптоты: x= –1” (это можно сделать, поскольку вертикальные асимптоты возникают в точках бесконечного разрыва). С новой строки наберите: “Коэффициенты наклонной асимптоты:”. Перейдите в режим командной строки и наберите:

> **k1:=limit(f/x, x=+infinity);**

$$k1:=1$$

> **b1:=limit(f-k1*x, x=+infinity);**

$$b1:=-3$$

```
> k2:=limit(f/x, x=-infinity);
```

$$k2 := 1$$

```
> b2:=limit(f-k2*x, x=-infinity);
```

$$b2 := -3$$

В этом случае коэффициенты наклонных асимптот при $x \to +\infty$ и $x \to -\infty$ оказались одинаковыми. Поэтому перейдите в текстовый режим и наберите “Уравнение наклонной асимптоты:”. Затем в новой строке перейдите в режим командной строки и наберите:

```
> ya=k1*x+b1;
```

$$ya = x - 3$$

В текстовом режиме наберите “Нахождение экстремумов”. В новой строке наберите команды:

```
> extrema(f,{},x,'s');s;
```

$$\{\frac{-256}{27}, 0\}$$

$$\{\{x = -4\}, \{x=0\}\}$$

Поскольку функция имеет разрыв, то при поиске максимума и минимума следует указать интервал, в который не должна входить точка разрыва.

```
> fmax:=maximize(f,x,x=-infinity..-2,location);
```

$$fmax := -\frac{256}{27}, \left\{\left[\{x = -4\}, -\frac{256}{27}\right]\right\}$$

```
> fmin:=minimize(f,x,x=-1/2..infinity,location);
```

$$fmin := 0, \{[\{x = 0\}, 0]\}$$

В текстовом режиме наберите результат исследования в виде:

“Максимум в точке (–4, –256/27); минимум в точке (0, 0).”

2. Построить график функции $y = \mathrm{arctg}(x^2)$ и ее асимптоту, указать координаты точек экстремума. Оформление каждого этапа исследования функции проделать также как и при выполнении предыдущего задания.

```
> restart: y:=arctan(x^2):
> iscont(y, x=-infinity..infinity);
```

$$true$$

```
> k1:=limit(y/x, x=-infinity);
```

$$k1 := 0$$

```
> k2:=limit(y/x, x=+infinity);
```

$$k2 := 0$$

```
> b1:=limit(y-k1*x, x=-infinity);
```

$$b1 := \frac{1}{2}\pi$$

```
> b2:=limit(y-k1*x, x=+infinity);
```

$$b2 := \frac{1}{2}\pi$$

```
> yh:=b1;
```

$$yh := \frac{1}{2}\pi$$

```
> extrema(y,{},x,'s');s;
```

$$\{0\}$$

$$\{\{x=0\}\}$$

```
> ymax:=maximize(y,x); ymin:=minimize(y,x);
```

$$ymax := \frac{1}{2}\pi$$

$$ymin := 0$$

```
> with(plots): yy:=convert(y,string):
> p1:=plot(y,x=-5..5, linestyle=1, thickness=3, color=BLACK):
> p2:=plot(yh,x=-5..5, linestyle=1,thickness=1):
> t1:=textplot([0.2,1.7,"Асимптота:"], font=[TIMES, BOLD, 10],
  align=RIGHT):
> t2:=textplot([3.1,1.7,"y=Pi/2"],font=[TIMES, ITALIC, 10],
  align=RIGHT):
> t3:=textplot([0.1,-0.2,"min:(0,0)"], align=RIGHT):
> t4:=textplot([2,1,yy], font=[TIMES, ITALIC,10],
align=RIGHT):
> display([p1,p2,t1,t2,t3,t4]);
```



§4. Локальные и условные экстремумы функций многих переменных

Для исследования функции на локальный и условный экстремум используется команда из стандартной библиотеки **`extrema(f,{cond},{x,y,…},'s')`**, где **`cond`** – ограничения для поиска условного экстремума, которые записываются в виде равенств. После ограничений в фигурных скобках указываются все переменные, от которых зависит функция **`f`**, а затем в кавычках записывается **`s`** – имя переменной, которой будут присвоены координаты точек экстремума. Если ограничений не указывать, то будет производиться поиск локального экстремума.

К сожалению, команда **`extrema`** выдает все критические точки, то есть и те, в которых экстремума нет. Отсеять недающие экстремума критические точки можно с помощью непосредственной подстановки этих точек в функцию, например, оператором **`subs`**.

Так же, как и для функции одной переменной, наибольшее и наименьшее значения функции многих переменных вычисляются командами **`maximize(f,{x1,…,xn},range)`**, и **`minimize(f,{x1,…,xn}, range)`**, где следует указывать после функции в фигурных скобках список всех переменных, от которых она зависит, а затем интервалы для каждой переменной, указывающие область поиска наибольшего и наименьшего значений.

Если требуется найти переменные, при которых линейная функция многих переменных имеет максимум (или минимум) при выполнении определенных ограничений, заданных в виде линейных равенств или неравенств, то следует использовать симплекс-метод. Для этого сначала необходимо загрузит пакет **simplex**, а затем воспользоваться командой **maximize** (или **minimize**), где теперь в качестве **range** можно указывать в фигурных скобках ограничительную систему неравенств. Пакет **simplex** предназначен для решения задач линейной оптимизации. После его загрузки команды **maximize** и **minimize** меняют свое действие. Теперь эти команды выдают координаты точек, при которых заданная линейная функция имеет максимум или минимум. При этом допускается дополнительная опция для поиска только неотрицательных решений **NONNEGATIVE**.

Задание 4

1. Найти экстремумы функции $f(x,y)=2x^4+y^4-x^2-2y^2$.

```
> restart:
> f:=2*x^4+y^4-x^2-2*y^2:
> extrema(f,{},{x,y},'s');s;
```

$$\{0, \frac{-9}{8}\}$$

$$\left\{\{x=0, y=-1\}, \{x=0, y=0\}, \{x=0, y=1\}, \left\{x=-\frac{1}{2}, y=-1\right\}, \left\{x=-\frac{1}{2}, y=0\right\}, \left\{x=-\frac{1}{2}, y=1\right\}, \left\{x=\frac{1}{2}, y=-1\right\}, \left\{x=\frac{1}{2}, y=0\right\}, \left\{x=\frac{1}{2}, y=1\right\}\right\}$$

Получилось всего два экстремума, поэтому очевидно, что f_{max}=0 и f_{min}=–9/8, причем максимум достигается в точке (0,0). Остальные критические точки следует проверить. В силу четности функции по обеим переменным, можно ограничиться проверкой критических точек только с положительными координатами.

```
> subs([x=1/2,y=1],f);
```

$$\frac{-9}{8}$$

```
> subs([x=1/2,y=0],f);
```

$$\frac{-1}{8}$$

```
> subs([x=0,y=1],f);
```

$$-1$$

Таким образом, функция имеет следующие локальные экстремумы: $f_{max}=f(0,0)=0$ и $f_{min}=f\left(\pm\frac{1}{2},\pm1\right)=f\left(\pm\frac{1}{2},\mp1\right)=-9/8$.

2. Найти наибольшее и наименьшее значения функции $f(x,y)=x^2+2xy-4x+8y$ в прямоугольнике x=0, y=0, x=1, y=2.

```
> restart:
> f:=x^2+2*x*y-4*x+8*y:
> maximize(f,x,y,x=0..1,y=0..2);
```

$$17$$

```
> minimize(f,x,y,x=0..1,y=0..2);
```

$$-3$$

Таким образом, функция имеет наибольшее значение f_{max}=17 и наименьшее значение f_{min}=–3.

3. Найти условные экстремумы функции $f(x,y,z)=xy+yz$ при $x^2+y^2=2$, $y+z=2$, $x>0$, $y>0$, $z>0$.

```
> restart: f:=x*y+y*z:
> assume(x>0);assume(y>0);assume(z>0);
> simplify(extrema(f,{x^2+y^2=2,y+z=2},{x,y,z},'s'));s;
```

$$\left\{2, -\frac{5}{2}-\frac{3}{2}\sqrt{3}\right\}$$

$$\left\{\{x\sim=-1, y\sim=1, z\sim=1\}, \{x\sim=1, y\sim=1, z\sim=1\}, \left\{x\sim=-\frac{1}{2}-\frac{1}{2}\sqrt{3}, y\sim=-\frac{1}{2}+\frac{1}{2}\sqrt{3}, z\sim=\frac{5}{2}-\frac{1}{2}\sqrt{3}\right\}, \left\{x\sim=-\frac{1}{2}+\frac{1}{2}\sqrt{3}, y\sim=-\frac{1}{2}-\frac{1}{2}\sqrt{3}, z\sim=\frac{5}{2}+\frac{1}{2}\sqrt{3}\right\}\right\}$$

Получили два экстремума и четыре экстремальных точки. В каких точках функция имеет экстремумы, можно определить подстановкой:

```
> subs(s[1],f);
```

$$0$$

```
> subs(s[2],f);
```

$$2$$

```
> simplify(subs(s[3],f));
```

$$-\frac{5}{2}+\frac{3}{2}\sqrt{3}$$

```
> simplify(subs(s[4],f));
```

$$-\frac{5}{2}-\frac{3}{2}\sqrt{3}$$

Таким образом, функция имеет следующие условные экстремумы: $f_{max}=f(1,1,1)=2$ и $f_{min}=f(-\frac{1}{2}+\frac{1}{2}\sqrt{3}, -\frac{1}{2}-\frac{1}{2}\sqrt{3}, \frac{5}{2}+\frac{1}{2}\sqrt{3}) = -\frac{5}{2}-\frac{3}{2}\sqrt{3}$

4. При каких значениях переменных функция $f(x,y,z)=-x+2y+3z$ имеет максимум, если требуется выполнение условий $x+2y-3z\le 4$, $5x-6y+7z\le 8$, $9x+10z\le 11$, а все переменные неотрицательные?

```
> restart: with(simplex):
> f:=-x+2*y+3*z:
> cond:={x+2*y-3*z<=4, 5*x-6*y+7*z<=8,
  9*x+10*z<=11}:
> maximize(f,cond,NONNEGATIVE );
```

$$\{x=0,\ y=\frac{73}{20},\ z=\frac{11}{10}\}$$

§5. Разложение в ряд и аппроксимация

Разложение функции в степенной ряд и ряд Тейлора.

Разложение функции $f(x)$ в степенной ряд в окрестности точки x_0

$$f(x)=\sum_{n=0}^{\infty} a_n (x-x_0)^n = a_0 + a_1(x-x_0) + a_2(x-x_0)^2 + \ldots + a_n(x-x_0)^n$$

осуществляется командой **`series(f(x), x=x0, n)`**, где **`x0`** – точка, в окрестности которой производится разложение, **`n`** – порядок разложения.

Аналогичного действия команда **`taylor(f(x), x=a, n)`** раскладывает функции **`f(x)`** в окрестности точки **`x=a`** по формуле Тейлора:

$$f(a)+\frac{f'(a)}{1!}(x-a)+\frac{f''(a)}{2!}(x-a)^2+\ldots+\frac{f^{(n)}(a)}{n!}(x-a)^n+\ldots$$

Команды **`series`** и **`taylor`** выдают результат, имеющий тип series. Для того, чтобы иметь возможность дальнейшей работы с полученным разложением, его следует преобразовать в полином с помощью команды **`convert(%,polynom)`**.

Функцию многих переменных $f(x_1,\ldots,x_n)$ можно разложить в ряд Тейлора по набору переменных $(x_1,\ldots,x_n)$ в окрестности точки $(a_1,\ldots,a_n)$ до порядка n с помощью команды **`mtaylor(f(x), [x1,…,xn], n)`**.

Задание 5

1. Разложить в степенной ряд $f(x)=e^{-x}\sqrt{x+1}$ в окрестности x_0=0, удерживая 5 первых членов.

```
> f(x)=series(exp(-x)*sqrt(x+1), x=0, 5);
```

$$f(x)=1-\frac{1}{2}x-\frac{1}{8}x^2+\frac{13}{48}x^3-\frac{79}{384}x^4+O(x^5)$$

2. Построить на одном рисунке графики интеграла ошибок $\operatorname{erf}(x)=\frac{2}{\sqrt{\pi}}\int\limits_0^x e^{-t^2}dt$ и его разложения в ряд Тейлора в окрестности нуля.

```
> taylor(erf(x),x,8): p:=convert(%,polynom);
```

$$p:=2\frac{x}{\sqrt{\pi}}-\frac{2}{3}\frac{x^3}{\sqrt{\pi}}+\frac{1}{5}\frac{x^5}{\sqrt{\pi}}-\frac{1}{21}\frac{x^7}{\sqrt{\pi}}$$

```
> plot([erf(x),p],x=-2..2,thickness=[2,2],
  linestyle=[1,3], color=[red,green]);
```



Пунктирной линей изображен график ряда Тейлора, а сплошной – самой функции.

3.Разложить функцию двух переменных $f(x,y)=\sin(x^2+y^2)$ в ряд Тейлора в окрестности точки (0, 0) до 6-ого порядка.

```
> f=mtaylor(sin(x^2+y^2), [x=0,y=0], 7);
```

$$f=x^2+y^2-\frac{1}{6}x^6-\frac{1}{2}y^2x^4-\frac{1}{2}y^4x^2-\frac{1}{6}y^6$$